

## 9/8.14.1

# Comptez vos pages et lignes de listing

---

Vous avez probablement apprécié notre programme permettant le formatage de listing, proposé en Partie 9, chapitre 8.14.

Ce programme possède une particularité très intéressante, pour éviter les chevauchements de lignes entre deux pages de papier listing, ou pour vérifier vos programmes par rapport à ceux que nous vous proposons, en paramétrant le nombre de caractères par ligne.

Il s'avère parfois, avec certains paramétrages, que le nombre de pages dépasse le nombre de feuilles encore disponibles dans votre carton de feuilles.

Parfois encore, lorsque vous désirez partager vos propres programmes avec un ami, si l'échange s'effectue par le poste, pour un même poids d'encre, vous risquez d'avoir un nombre de copies impressionnant et un prix multiplié d'autant.

Aussi, nous vous proposons ici un programme complémentaire au formateur de listing, qui vous permettra de compter le nombre de lignes et de pages pour un format d'impression donné. Lorsque ce nombre vous satisfaira, vous pourrez lancer l'impression.

### ETUDE DU PROGRAMME

#### L'algorithme général


Nous vous proposons ici l'algorithme général du programme, qui insiste essentiellement sur le traitement des lignes du fichier programme. Nous verrons plus en détail le fonctionnement du programme en précisant plus loin l'algorithme.


- DEBUT
  - Acquérir les données de paramétrage
    - CO
      - nom du programme
      - nombre de lignes par page
      - nombre de caractères par ligne
    - FINCO

- Ouvrir le fichier concerné en lecture
- TANTQUE la fin du fichier n'est pas atteinte
  - Lire une ligne du fichier
  - TANTQUE le nombre de lignes par page n'est pas atteint
    - Calculer le nombre de lignes sur le papier que prend la ligne de programme
    - Remettre à jour le nombre de pages
    - Afficher la ligne de programme
- FINTANTQUE
- Augmenter le nombre de pages de 1
- Afficher le nombre de lignes
  - Afficher le nombre de pages
- FINTANTQUE
- Fermer le fichier
- FIN

#### **L'ALGORIGRAMME**

Nous vous proposons page suivante l'algorithme détaillant le fonctionnement général du programme de calcul de pages de listing.





**LE PROGRAMME**

De conception simple, nous vous proposons un programme de quelques dizaines de lignes, numérotées de dix en dix pour vous permettre une saisie en mode AUTO.

```

10 REM *****
20 REM **                               **
30 REM ** UTILITAIRE DE COMPTAGE       **
40 REM ** DE LIGNES D'UN PROGRAMME    **
50 REM ** BASIC SAUVE SOUS FORME      **
60 REM ** ASCII POUR DETERMINER LE    **
70 REM ** NOMBRE DE FEUILLES DE       **
80 REM ** LISTING NECESSAIRES         **
90 REM **                               **
100 REM *****
110 REM
120 MODE 2
130 WINDOW #1,1,80,1,7
140 REM
150 REM ** ENTREE DES PARAMETRES **
160 REM
170 PRINT #1,"Le programme a traite doit
avoir,ete sauve sous forme ASCII !"
180 PRINT #1
190 PRINT #1,"Entrez le nom du programme
: ";
200 INPUT #1,nomprogramme$
210 REM
220 PRINT #1,"Nombre de caracteres par l
igne : ";
230 INPUT #1,caractereligne%
240 PRINT #1,"Nombre de ligne par page
: ";
250 INPUT #1,lignepage%
260 REM
270 REM ** OUVERTURE FENETRES **
280 REM
290 IF caractereligne% > 80 THEN WINDOW
#0,1,80,8,25: GOTO 340
300 WINDOW #0,1,caractereligne%,8,25
310 REM
320 REM ** FENETRES AFFICHAGE PARAMETRES
**
330 REM
340 CLS#1
350 LOCATE #1,23,3
360 PRINT #1,"Traitement de : ";
370 PRINT #1,CHR$(24) + nomprogramme$ +
CHR$(24)

```

```
380 LOCATE #1,10,5
390 PRINT #1,"Nombre de lignes"
400 LOCATE #1,45,5
410 PRINT #1,"Nombre de pages"
420 REM
430 REM ** OUVERTURE FICHER **
440 REM
450 OPENIN nomprogramme$
460 REM
470 REM ** TRAITEMENT **
480 REM
490 WHILE NOT EOF
500     LINE INPUT #9,ligne$
510     longueurligne% = LEN(ligne$)
520     nombreligne% = nombreligne% + INT
        (longueurligne% / (caractereligne% + 1))
        + 1
530     nombrepageancien% = nombrepage%
540     nombrepage% = INT ((nombreligne%
- 1) / lignepage%) + 1
550     IF nombrepageancien% = nombrepage
% THEN GOTO 620
560     PRINT
570     PRINT STRING$(caractereligne%,CHR
$(45));
580     PRINT "Page No";
590     PRINT nombrepage%
600     PRINT
610     REM
620     PRINT ligne$
630     LOCATE #1,16,6
640     PRINT #1,CHR$(24) + STR$(nombreli
gne%) + CHR$(32) + CHR$(24)
650     LOCATE #1,51,6
660     PRINT #1,CHR$(24) + STR$(nombrepa
ge%) + CHR$(32) + CHR$(24)
670 WEND
680 REM
690 REM ** FIN TRAITEMENT FERMETURE FICH
IER **
700 REM
710 CLOSEIN
720 REM
730 REM ** RESTITURION FENETRE 0 **
740 REM
750 WINDOW #0,1,80,1,25
760 REM
770 LOCATE 1,25
780 PRINT
790 END
```

Lignes 120 et 130 : effacement de l'écran et définition de la fenêtre numéro 1 (#1) pour instaurer le dialogue et afficher les caractéristiques du listing.

Ligne 170 : remarque importante.

Lignes 180 à 250 : acquisition du nom du programme et des caractéristiques de formattage du listing.

Lignes 290 et 300 : initialisation de la fenêtre numéro zéro (#0) au format en largeur du listing. Si le nombre de caractères par ligne est supérieur à 80, la fenêtre est limitée à 80 à cause de la définition d'affichage sur l'écran en mode 2.

Lignes 340 à 410 : affichage des zones paramètres de formattage, et rappel du nom du fichier, en fenêtre #1.

Lignes 490 et 670 : boucle de lecture du fichier.

Ligne 500 : lecture d'une ligne dans le fichier.

Ligne 510 : calcul de la longueur de la ligne.

Ligne 520 : calcul du nombre de lignes déjà lues et formattées pour l'impression.

Ligne 530 : sauvegarde nombre de pages actuel.

Ligne 540 : calcul du nouveau nombre de pages.

Lignes 550 à 590 : affichage ou non d'une nouvelle page, si le nouveau nombre de pages est différent ou non de l'ancien.

Ligne 620 : affichage de la ligne en fenêtre #0.

Lignes 630 à 660 : affichage des caractéristiques nombre de lignes et nombre de pages.

Lignes 710 à Fin : fermeture du fichier et restitution de la fenêtre zéro à sa taille originale.

### UTILISATION

Après lancement, le programme vous demande le nom du fichier à traiter (ce pourrait être un listing, comme tout autre fichier).

Une remarque importante : le fichier doit avoir été sauvegardé sous forme ASCII, donc par la commande suivante, si c'est un programme :

**SAVE "MONPROG.ASC", A**

Vous entrerez ensuite le nombre de caractères par ligne, puis le nombre de lignes par page.

S'affichent ensuite, les lignes formatées du listing dans la fenêtre inférieure de l'écran, avec une séparation et une numérotation des pages.

Dans la fenêtre supérieure est rappelé le nom du fichier, et mis à jour le nombre de lignes formatées ainsi que le nombre de feuilles de papier qu'il vous faudra pour imprimer le listing au format désiré.

La figure ci-dessous vous donne un exemple de traitement du programme de gestion de tour de contrôle qu vous aurez plaisir à découvrir dans votre ouvrage « Comment exploiter toutes les possibilités... ».

Si, en fin de traitement, ces nombres ne vous conviennent pas, il vous sera possible de relancer le programme pour un traitement avec des paramètres différents.

```

                                Traitement de : ALUCONTROL
                                Nombre de lignes      Nombre de pages
                                2250                2250
2250 WEND
2260 a$ = UPPER$(a$)
2270 REM
2280 REM traitement ordre valide
2290 REM
2300 flag% = 0
2310 IF a$ = "A" THEN GOSUB 2770 : REM Atterissage demande
2320 IF a$ = "M" THEN GOSUB 3870 : REM Mouvement d'avion dem
ande
2330 IF a$ = "P" THEN GOSUB 4590 : REM Pas d'action - Attent
e
-----
Page No 5
2340 IF a$ = "Z" THEN GOSUB 5140 : REM Demande d'etat de zon
e
2350 IF a$ = "D" THEN GOSUB 5470 : REM Decollage avion

```